# Maintainer: Thorsten Töpper <atsutane-tu@freethoughts.de>

pkgname=perl-anyevent-i3
pkgver=0.17
pkgrel=9
pkgdesc="Communicate with the i3 window manager"
arch=('any')
url='https://search.cpan.org/~mstplbg/AnyEvent-I3/'
license=('GPL')
depends=('perl' 'perl-anyevent' 'perl-json-xs')
options=('!emptydirs')
source=(https://search.cpan.org/CPAN/authors/id/M/MS/MSTPLBG/AnyEvent-I3-${pkgver}.tar.gz)
sha256sums=('5382c984c9f138395f29f0c00af81aa0c8f4b765582055c73ede4b13f04a6d63')

build() {
  cd AnyEvent-I3-${pkgver}
  PERL_USE_UNSAFE_INC=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd AnyEvent-I3-${pkgver}
  make test
}

package() {
  cd AnyEvent-I3-${pkgver}
  make install DESTDIR="${pkgdir}"
}
